

```

; =====
;
;   Simulation of pull-test for a grouted cable anchor
;
; =====
set logfile pull1.log
set log on

new
title
Pull Test for Grouted Cable Anchor (no external confinement, fric=0)

; =====
; Create a single rock block and set its material properties.
;
gen zone brick size 4 7 4
mod el
ini x mul .1
ini y mul .1
ini z mul .1
;
pro bulk 5e9 she 3e9

; =====
; Create a single cable and set its associated properties
;
sel cable id=1 begin=(0.2, 0.0, 0.2) end=(0.2, 0.5, 0.2) nseg=10
sel cable prop Xcarea=181e-6 Emod=98.6e9 YTension=0.232e6 &
               GR_K=1.12e7 GR_Coh=1.75e5 GR_Fric=0.0 GR_Per=7.85e-2

; =====
; Fix free end of rock block and apply velocity to cable end
;
def initval
    pull_vel = -1.0e-6
end
initval
;
fix y range y=(-0.01 0.01)
cycle 1 ; force the local-nodal systems of all nodes associated with
        ; cable to be updated
sel node FIX y range y=(-0.01 0.01)
sel node INIT YVel=pull_vel range y=(-0.01 0.01)

```

```

; =====
; Create FISH function for monitoring total axial force built up
; in the rock per unit cable length [force_len] and
; displacement of cable-end [disp_end].
;
def force_len
;
  _sum = 0.0
  gpp = gp_head
  loop while gpp # null
    if gp_ypos(gpp) <= 0.05 then
      _sum = _sum + gp_yfunbal(gpp)
    end_if
    gpp = gp_next(gpp)
  end_loop
  force_len = _sum / 0.5
  disp_end = step * (-1.0)*pull_vel
end

; =====
; Set up histories for monitoring model behavior
;
history nstep=10
hist id=1 unbal
hist id=10 force_len
hist id=20 sel node      ydisp id=1
hist id=100 sel cableSEL force id=1
hist id=200 sel cableSEL force id=3
hist id=300 sel cableSEL force id=5
hist id=400 sel cableSEL force id=7
hist id=500 sel cableSEL force id=9
hist id=600 sel cableSEL force id=10

; =====
; Apply velocity to achieve total displacement of 2.0 cm
;
cycle 10000
sav pull11.sav
cycle 6500
sav pull12.sav
cycle 1000
sav pull13.sav
cycle 2500

```

```
; =====  
; Save file and turn off logfile  
;  
sav pull14.sav  
set log off  
  
; =====  
return
```